
ABSTRACT. This is the second paper in a series presenting case studiesin modern large-scale constrained
optimization [9]. In this paper, we consider the shape of a hanging chain, which, in equilibrium, minimizes
the potential energy of the chain. In addition to the tutorial aspects of this paper, we also emphasize the
importance of certain modeling issues such as convex vs. nonconvex formulations of given problem. We will
present several models of the problem and demonstrate differences in the number of iterations and solution
time.

1. INTRODUCTION

The hanging chain or catenary problem (the word “catenary” comes from the Latin word “catena” mean-
ing chain) was first posed in the Acta Eruditorium in May 1690 by Jacob Bernoulli as follows: ”To find
the curve assumed by a loose string hung freely from two fixed points”. Earlier, Galileo mistakenly con-
jectured that the curve was a parabola. Later Joachim Jung proved that the curve cannot be a parabola but
without presenting any solution of the real curve. In June 1691 there were three solutions published, from
Leibniz, Huygens and Johann Bernoulli brother of Jacob. Even though these mathematicians approached
this problem in three different ways they concluded that thecurve was the hyperbolic cosine, which then
came to be known as the catenary.

In more recent times, the catenary curve has come to play an important role in civil engineering. The
solution of the catenary problem provides the starting point for consideration of the effects on a suspended
cable of extraneous applied forces such as arising from the live loads on a practical suspension bridge.

However, in the real world, the problem of finding an optimal construction shape is more complicated
than the original catenary problem. Jacob Bernoulli assumed that the string is flexible and of uniform
cross-section, which implies that every segment of equal length has equal mass. This assumption is too
restrictive for modern engineering. Moreover, in many practical suspension bridges the total weight of
the bridge, instead of being uniformly distributed along the cable, is actually more uniformly distributed
across the bridge span. In this case, the shape of the cable iscloser to a parabola than a catenary.

Key words and phrases.shape optimization, optimal design, constrained optimization, catenary problem.

The purpose of this paper is to demonstrate how a general chain problem can be modeled as an opti-
mization problem and solved by widely available optimization tools. Throughout the paper we present
several optimization models and demonstrate how to solve them using optimization software. We express
these models in theAMPL modeling language [5] for two reasons. First, this languageis used as a com-
mon mechanism for conveying optimization problems. So a reader can copy and use the models to try the
majority available solvers. On the other hand, anAMPL model is easily readable even for one not familiar
with exactAMPL syntax. Therefore one can compare different models.

We selectedLOQO [6, 7, 8] as a nonlinear optimization solver, which implements an interior point
method (IPM) for general nonlinear optimization and adequately serves our needs. The performance of
other solvers for the catenary problem is studied in [3].

This paper is intended to be a tutorial on optimization. We emphasize the importance of proper model-
ing.

The main point we wish to illustrate is that any given physical problem can have multiple equivalent
mathematical formulations some of which are numerically tractable while others are not. More specifi-
cally, we focus on convex vs. nonconvex formulations of a problem. An optimization problem is convex
if it can be expressed as:

min f(x),

subject to

ci(x) ≤ 0, i = 1, . . . , m,

aj(x) = 0, j = 1, . . . , p,

wheref(x), ci(x), i = 1, . . . , m are convex andaj(x), j = 1, . . . , p are affine functions. Convex opti-
mization problems are much more numerically tractable thannonconvex ones.

The paper is organized as follows. In the next section, we consider the catenary problem with the
mass uniformly distributed along its length. We consider two modeling approaches. The first one leads
to the solution of a nonconvex formulation of the catenary problem in the direct formy(x), x ∈ [xa, xb].
The second model leads to the solution of a convex formulation of the problem in a parametric form
(x(t), y(t)), t ∈ [0, l].

Section 3 considers the case when the mass of a chain is uniformly distributed along the horizon. Again,
as in the previous section, two modeling approaches are considered leading toy(x) and(x(t), y(t)) re-
spectively.

2. CHAIN PROBLEM, DISCRETIZATION ALONG THE HORIZON.

The common formulation of the variational problem for finding the shape of a heavy chain of length
l with uniformly distributed mass along the chain is to find a functiony(x) that minimizes the potential
energy

(1) min

∫ xb

xa

y

√

1 + y′2dx,

FIGURE 1. Catenary

wherey is height and
√

1 + y′2dx is proportional to mass, subject to constraints:

(2)
∫ xb

xa

√

1 + y′2dx = l,

and

(3) y(xa) = ya, y(xb) = yb.

The solution to this problem is

(4) y(x) = C cosh
x + C1

C
+ C2,

where the values ofC, C1 and C2 are determined by conditions (2)-(3) ([1, 2]). Figure 1 shows the
graphical representation ofy(x).

2.1. Discretization. In [9], we considered two simple methods for discretizing calculus of variations
problems: the trapezoidal discretization and the midpointdiscretization. We showed that the trapezoidal
method, which seems to be the most widely used method, is morelikely to exhibit nonuniqueness of the
solution when there are extra degrees of freedom over which one is optimizing. In this paper, we shall use
the midpoint discretization in all of the examples.

If xa = x0 < x1 < · · · < xN−1 < xN = xb is the uniform discretization of segment[xa, xb], the
midpoint method approximates integrals as follows

∫ b

a

f(x)dx ≈
N−1
∑

i=0

f(xi+ 1

2

)(xi+1 − xi), where xi+ 1

2

=
xi + xi+1

2
.

param N := 200;
param d := 5;
param ax := 0; param ay := 0;
param bx := d; param by := 0;
param dx := (bx - ax)/N;
param x {j in 0..N} :=
ax * (1-j/N) + bx * j/N;
param l := 2 * d; param g := 9.8;
param m {0..N}, default 1;

var y {0..N};
var ydot {j in 1..N};

minimize energy:
sum {j in 1..N}
dx * m[j] * g* sqrt(1+ydot[j]ˆ2) *
(y[j]+y[j-1])/2;

s.t. eqn {j in 1..N}:
y[j] = y[j-1] + dx * ydot[j];
s.t. length:
sum {j in 1..N} sqrt(1+ydot[j]ˆ2) * dx = l;
s.t. leftfixed: y[0] = ay;
s.t. rightfixed: y[N] = by;

let {j in 0..N}
y[j] := (j/N) * by + (1-j/N) * ay;
let {j in 1..N}
ydot[j] := (y[j]-y[j-1])/dx;
solve;

FIGURE 2. TheAMPL modelmid.mod

Also, the method approximates the derivatives as follows

y′(xj+ 1

2

) =
yj+1 − yj

xj+1 − xj

, j = 0, 1, ..., N − 1

We avoid creating extra variablesy(xj+ 1

2

) in the middle of discretization intervals by approximating
y(xj+ 1

2

) ≈ 0.5(y(xj+1)+y(xj)). The function used to approximate integral (1) isf(xi+ 1

2

) = 0.5[y(xj+1)+

y(xj)]
√

1 + y′(xj+ 1

2

)2. The midpoint discretization of problem (1)-(3) in theAMPL modeling language is

shown in Figure 2.
We solved this problem withLOQO for various values of discretization parameterN. Table 1 shows the

behavior of the solver. In this and all following tables we show the solution time in seconds, infeasibility
or accuracy of the solution and the number of iterations.

N = 100 N = 200 N = 400 N = 800

time 1.41 14.68 248.02
obj. val = -223.2416282 -223.2449622 -223.2457957

infeasibility = 2.3e-11 4.9e-11 1.0e-11
steps 50 90 199 ≥500

TABLE 1. Chain model, discretized along the horizon. The number ofiterations and solu-
tion time grow significantly with the increase of the number of discretization points.

2.2. Catenary problem – discretization along the chain.The discretized constraint (2) can be a source
of numerical problems for an interior point algorithm. We found that parametrization of the chain along
itself can lead to a model that is simpler to attack both mathematically and numerically.

Let’s consider again a chain with uniformly distributed mass along the cable. Parametrizing the curve
using arclengtht, implying dx2 + dy2 = dt2, we look for (x(t), y(t)), t ∈ [0, l], which minimizes the
potential energy

(5) min

∫ l

0

m(t)y(t)dt,

such that

(6) ẋ2 + ẏ2 = 1,

and

(7) x(0) = xa, y(0) = ya, x(l) = xb, y(l) = yb.

wherem(t) = m = const,ẋ = dx/dt, ẏ = dy/dt. The following proposition shows that the solution of
problem (5)-(7) coincides with that of (1)-(3).

Proposition 1. Let (x(t), y(t)), t ∈ [0, l] be the solution of problems (5)-(7). The functionx(t) is strictly
monotone and hence invertible. Furthermore,

y
(

t−1(x)
)

= C cosh
x − C1

C
+ C2,

where the values ofC, C1 andC2 are determined by conditions (6)-(7).

Proof. We can assume thatm = 1. Also, we can assume thatẋ 6= 0, t ∈ [0, l]. Otherwise it would
contradict the laws of physics: the horizontal components of tensionsT1 andT applied to any portion of
the chain SR should be equilibrated by the second Newton’s law (see Figure 3). Therefore if(x(t), y(t))
is the solution of problem (5)-(7) there exists a multiplierλ = λ(t), t ∈ [0, l] such that(x(t), y(t)) is an
extremal without side condition for the functional (see [1])

(8) min

∫ l

0

(

y + λ(ẋ2 + ẏ2 − 1)
)

dt

The Euler equations for problem (8) are
d
dt

2λẋ = 0,
d
dt

2λẏ = 1.

Integrating the first equation we obtain2λẋ = C. Keeping in mind thaṫx 6= 0, t ∈ [0, l] we can express
λ = C/(2ẋ) and substitute into the second equation, which becomes

(9) C
d

dt

ẏ

ẋ
= 1.

Since we look for a solution of the formy(x), we have

(10) C
d

dt

ẏ

ẋ
= C

d

dt
y′ = Cy′′ẋ

(Note: we use “dot” to denote differentiation with respect to t and “prime” to denote differentiation with
respect tox). Also (6) turns to

ẋ2 + ẏ2 = ẋ2 + (y′ẋ)2 = 1,

FIGURE 3. Catenary

which is equivalent to

(11) ẋ =
1

√

1 + y′2
,

under the assumption thatẋ > 0, t ∈ [0, l]. Therefore combining (9), (10), and (11) we obtain the following
ODE

Cy′′

√

1 + y′2
= 1,

which can be rewritten as

C
dy′

√

1 + y′2
= dx.

After integrating this ODE we have
x = C sinh−1 y′ + C1,

which is equivalent to

y′ = sinh

(

x − C1

C

)

.

Finally after another integration we obtain

y(x) = C cosh
x − C1

C
+ C2.

This proves the proposition.
Figure 4 presents the discretized problem (5)-(7). This optimization problem has N constraints:

s.t. link {j in 1..N}: (x[j]-x[j-1])ˆ2+(y[j]-y[j-1])ˆ2 = ( l/N)ˆ2;

param N := 200;
param d:=5;
param ax := 0; param ay := 0;
param bx := d; param by := 0;
param l :=2 * d; param g := 9.8;
param m {1..N-1}, default 1;

var x {0..N};
var y {0..N};

minimize energy:
(l/N) * sum {j in 1..N-1} m[j] * g* y[j];

s.t. link {j in 1..N}:
(x[j]-x[j-1])ˆ2+(y[j]-y[j-1])ˆ2
= (l/N)ˆ2;
s.t. xleftfixed: x[0] = ax;
s.t. yleftfixed: y[0] = ay;
s.t. xrightfixed: x[N] = bx;
s.t. yrightfixed: y[N] = by;

let {j in 0..N} x[j] :=
(j/N) * b_x + (1-j/N) * a_x;
let {j in 0..N} y[j] :=
(j/N) * b_y + (1-j/N) * a_y;

solve;

FIGURE 4. TheAMPL modelchain.mod

which are a potential source of numerical problems. Indeed,LOQO fails to solve this problem even with
N = 100.

However, the formulation can be simplified drastically by relaxing the equality in the “link” constraints
into inequalities:

s.t. link {j in 1..N}: (x[j]-x[j-1])ˆ2+(y[j]-y[j-1])ˆ2 <= (l/N)ˆ2;

It can be shown by contradiction that an optimal solution to the problem in Figure 4 with the relaxed “link”
constraints satisfies the constraints as equalities. Indeed, suppose that one or more “link” constraints are
strict inequalities for an optimal solution. Then by rotating subchains around fixed points A and B (see
Figure 1) or by lowering the sub chain preserving strict inequality we can decrease the potential energy of
the whole chain, which contradicts optimality.

At the same time, the relaxed problem is convex with a linear objective function and easily solvable by
any nonlinear optimization code in particularLOQO. Table 2 displays solution statistics.

N = 100 N = 200 N = 400 N = 800

time 0.12 0.24 0.53 1.10
obj. val = -223.2381141 -223.2440837 -223.2455761 -223.2459492

infeasibility = 1.4e-11 3.0e-12 2.6e-13 2.6e-13
steps 22 24 27 28

TABLE 2. Parametrized chain. The number of iterations practically independent of the
number of dicretization points. The table demonstrates a drastic improvement of the solu-
tion time comparing to the results in Table 1.

If the mass function is not a constant along the chain but is instead any general known functionm(t)
then it is still easy to obtain a solution of a hanging chain problem. For example, let the mass of each node
equal 1 except for three special nodes: the center node and the two nodes one quarter of the length away
from both end points. This specification is modeled by makingthe following addition to theAMPL code
in Figure 4:

let {j in 1..3} m[int((N-1) * j/4)] := 0.3 * N;

FIGURE 5. Not uniformly distributed mass function

The resulting shape of the chain is shown in Figure 5 while Table 3 presents the statistics of the solver’s
behavior.

N = 100 N = 200 N = 400 N = 800

time 0.21 0.35 0.80 1.80
obj. val = -474.7789365 -479.2553063 -481.4659418 -482.5641514

infeasibility = 1.1e-11 7.8e-12 3.0e-11 2.4e-11
steps 36 36 42 46

TABLE 3. The shape of the hanging chain with not uniformly distributed but known mass
function can be easily found.

2.3. Lessons.Hanging chain problems are convex and therefore easy to solve if the model representation
is also convex. It is apparently not possible to make a convexmodel when parametrizing alongx. But it is
easy to do so when parametrizing along the chain. Indeed, theconvexity is created when constraint (5) is
relaxed to the inequality without changing the optimal solution. However, if the model is discretized along
the horizon, the relaxation of constraint (2) to inequalitygives a model that is not equivalent and results in
an incorrect solution (see Figure 6). The length of an extremal chain turns out to be less thanl, and thus
we cannot employ this trick for problem (1)-(3). Hence, the parametrized model (5)-(7) is more suitable
for the computational analysis.

FIGURE 6. Relaxed model, discretized along the horizon

3. BRIDGE PROBLEM.

The hanging chain model can be the starting point for studying other problems of structural design in
civil engineering. For example, let’s consider the problemof identifying the shape of the cable that carries
a suspension bridge. In this case, the total weight of the bridge, instead of being uniformly distributed
along the cable, is actually more uniformly distributed across the bridge span. This weight distribution
makes the bridge model more difficult to analyze.

In this section, we consider the bridge model and show how it can solved usingLOQO.

3.1. Bridge problem, discretization along the horizon. Let us first try to build a bridge model dis-
cretized along the horizontal axis, which implies that the mass function is constant:m(x) = m for anyx.
The variational problem that reflects this situation could be formulated as follows:

(12) min

∫ xb

xa

y(x)dx,

such that

(13)
∫ xb

xa

√

1 + y′2dx = l,

with

(14) y(xa) = ya, y(xb) = yb.

Surprisingly, the solutiony(x) to this problem is an arc segment of a circle:

Proposition 2. If 0 < l1 ≤ l ≤ l2, for some constantsl1 and l2 then the solutiony(x), x ∈ [xa, xb] of
problem (12)-(14) satisfies

(x − c)2 + (y(x) − d)2 = r2,

wherec, d andr are determined by conditions (13)-(14).

Proof. If y(x) is the solution to problem (12)-(14) then there exists a constant r, such thaty(x) is an
ordinary extremal of the following functional [1]:

(15)
∫ xb

xa

(y + r

√

1 + y′2)dx.

The solutiony(x), x ∈ [xa, xb] satisfies the Euler equation

d

dx

ry′

√

1 + y′2
= 1,

which after the integration becomes
ry′

√

1 + y′2
= x − c,

wherec is a constant of integration.
After squaring the last expression we obtain

(ry′)2 = (x − c)2(1 + y′2).

Since we are looking for a minimum of (12) thensgn(y′) = sgn(x − c). Therefore we can rewrite the last
expression as

y′ =
x − c

√

r2 − (x − c)2
,

which after integration becomes
y − d =

√

r2 − (x − c)2,

and therefore
(x − c)2 + (y(x) − d)2 = r2.

The proof of the proposition is complete.
Let us find a solution of (12)-(14) numerically. Figure 7 shows the correspondingAMPL model. Note

that constraint (13) is relaxed to an inequality to make the problem convex:
s.t. length: sum {j in 1..N} sqrt(1+ydot[j]ˆ2) * dx <= l;

Again, it can be shown by contradiction that the optimal solution of problem in Figure 7 with the relaxed
“length” constraints satisfies these constraints as equalities. The solution of this problem is the semicircle
(see Figure 8). Table 4 presents the statistics of the solver’s behavior.

Remark 3.2. For values ofl within a certain rangel1 ≤ l ≤ l2, the solution of problem (12)-(14) is
an arc of a circle. However, if the length of a cable is big enough it is no longer possible to find a circle
satisfying conditions (13)-(14). Then the solution of problem (12)-(14) cannot be found from a class of
smooth functionsy(x). Theoretical analysis corresponding to this case is complicated. It is simpler to find
solutions numerically. For example, let us double the length of a chain:

param l := 2 * d* 2* atan(1).

param N := 200;
param d := 5;
param ax := 0; param ay := 0;
param bx := d; param by := 0;
param dx := (bx - ax)/N;
param x {j in 0..N} :=
ax * (1-j/N) + bx * j/N;
param l := d * 2* atan(1); param g := 9.8;
param m {0..N}, default 1;

var y {0..N};
var ydot {j in 1..N};

minimize energy:
sum {j in 1..N}
dx * m[j] * g* (y[j]+y[j-1])/2;

s.t. eqn {j in 1..N}:
y[j] = y[j-1] + dx * ydot[j];
s.t. length:
sum {j in 1..N} sqrt(1+ydot[j]ˆ2) * dx <= l;
s.t. leftfixed: y[0] = ay;
s.t. rightfixed: y[N] = by;

let {j in 0..N}
y[j] := (j/N) * by + (1-j/N) * ay;
let {j in 1..N}
ydot[j] := (y[j]-y[j-1])/dx;
solve;

FIGURE 7. TheAMPL modelfixedmass.mod

FIGURE 8. The solution of the model in Figure 7 is a circle.

The graph of the solution is shown in Figure 9 and the solver’sbehavior is shown in Table 5. Our pre-
sumption is that the curveAMNB in Figure 9 consists of circle segmentMN and two vertical segmentsAM

andNB.
Remark 3.3. The solution to problem (12)-(14) does not lead to a real suspension bridge model with

uniformly distributed mass along the horizon. Actually, making an assumption of a mass distributionm(x)

N = 100 N = 200 N = 400 N = 800

time 0.22 0.98 6.43 57.79
obj. val = -96.16397810 -96.19471719 -96.20546126 -96.2092295

infeasibility = 2.6e-11 4.4e-12 2.1e-11 1.1e-11
steps 20 19 20 21

TABLE 4. Fixed uniformly distributed mass function along the horizontal axis

N = 100 N = 200 N = 400 N = 800

time 0.26 1.30 8.07 70.39
obj. val = -286.7872655 -287.7015964 -288.1639555 -288.3974098

infeasibility = 5.1e-12 1.5e-11 9.0e-12 5.8e-12
steps 23 25 25 26

TABLE 5. Long chain solution can be obtained numerically

FIGURE 9. “Long circle”

in criteria (12) results in the appearance of horizontal external forces, which prevent redistribution of the
chain mass along the horizon.

To clarify, let us consider the situation in Figure 10. Suppose we have fixed vertical rods with sliding
frictionless rings of given mass and a weightless string, which goes through all the rings. This is exactly

FIGURE 10. Fixed distribution of mass

the case when the distribution of mass along the horizon is fixed. If the system rests in the state of minimal
potential energy there will be normal horizontal forces coming from the rods. However the shape of the
hanging chain is the result of just the gravity and tension forces.

3.2. Lessons.Problem (12)-(14) cannot handle the case when we want to find the shape of the a chain
with a given distribution along the horizonm(x) due to the following difficulty: we cannot use function
m(x) in the criteria (12) because it leads to the appearance of horizontal external forces and brings us to a
wrong solution. The next subsection resolves this paradox when the chain is parametrized along itself.

3.3. Bridge model, discretization along the chain’s length.In this section we consider the situation
where the mass of the bridge is distributed along the horizonwhile the chain is parametrized along its
length(x(t), y(t)), t ∈ [0, l].

In the minimized integral criteria we cannot make assumptions on the mass distributions along the
horizon (see Remark 3.3). Therefore we assume that mass distribution depends only ont. In the following,
we consider again model (5)-(7) with a general mass distribution functionm(t)

(5) min

∫ l

0

m(t)y(t)dt,

such that

(6) ẋ2 + ẏ2 = 1,

and

(7) x(0) = xa, y(0) = ya, x(l) = xb, y(l) = yb.

For any distribution functionm(t) there is a solution to problem (5)-(7). However we are interested in
the specific case where the mass is uniformly distributed along the horizon when the minimum is found.
The following proposition proves that only the parabola canbe such a solution.

Proposition 3. Let (x(t), y(t)), t ∈ [0, l] be the solution to problem (5)-(7) whose mass is uniformly
distributed along the horizon, then

(16) y(x) = C1x
2 + C2x + C3,

where the values ofC1 > 0, C2 andC3 are determined by conditions (6)-(7).

Proof. We can assume thatẋ 6= 0, t ∈ [0, l], see (Proposition 3.1). Therefore if vector function
(x(t), y(t)) is the solution of problems (5)-(7) then there exists a multiplier λ = λ(t), t ∈ [0, l] such that
(x(t), y(t)) is an extremal without side condition for the functional

(17)
∫ l

0

(

m(t)y(t) + λ(t)(ẋ2 + ẏ2 − 1)
)

dt,

The Euler equations for problem (17) are

d
dt

2λẋ = 0,
d
dt

2λẏ = m(t).

Integrating the first equation we obtain2λẋ = C. Keeping in mind thaṫx > 0, t ∈ [0, l] we can express
λ = C/(2ẋ) and substitute to the second equation, which becomes

(18) C
d

dt

ẏ

ẋ
= m(t).

Since we look for the solution with uniformly distributed mass along the horizon the following relation
betweenm(t) andm(x) holds

(19) m(t) =
m(x)

√

1 + y′2
= m(x)ẋ = mẋ,

wherem = M/|xa − xb| andM is the mass of whole chain. The second equality follows from (11).
Therefore taking into account (10) and (19) we can rewrite (18) as follows

Cy′′ẋ = mẋ.

Sinceẋ > 0 we have

y′′ =
C1

2
, C1 =

2m

C
> 0,

which after integrating two times becomes (16). This provesthe proposition.
Remark 3.5. If instead of (19) one considersm(t) = m, t ∈ [0, l] for ordinary differential equation

(18), one can verify thaty(x) is a hyperbolic cosine (4). Therefore equations (18) and (6)can be treated
as general necessary conditions for problem (5)-(7) for anygiven mass functionm(t).



CASE STUDIES IN OPTIMIZATION: CATENARY PROBLEM 15

Although there is no explicit optimization criteria in (6),(18), we still argue that the solution to this
system can be obtained easily by using the solver. Let us solve this problem numerically for the case when
the mass is distributed uniformly along the horizon

(20) C
d

dt

ẏ

ẋ
= m(t),

(21) ẋ2 + ẏ2 = 1,

(22) x(0) = xa, y(0) = ya, x(l) = xb, y(l) = yb.

One possible way is to discretize (20), (21) without objective function subject to fixed end constraints
(22). This approach can lead to numerical problems, since the problem becomes nonlinear and nonconvex.
Instead we will try to ”convexify” the problem to make it moremanageable. Again we relax constraint
(21) to inequality and introduce a linear objective function, which guarantees that a solution will satisfy
constraint (21) as equality.

In order to simplify constraint (20) we integrate it

(23) C
ẏ

ẋ
=

∫ t

0

m(ξ)dξ − E = M(t) − E,

where a new variableM(t̄) =
∫ t̄

0
m(ξ)dξ is a total mass of a chain segment whent ∈ [0, t̄].

DiscretizingM(t) leads to the following

(24) Mt+1 = Mt +

∫ t+1

t

m(ξ)dξ

If mass is distributed uniformly along the horizon then discretized (24) becomes
M[t+1] = M[t] + m * (x[t+1] - x[t]),

while discretized (23) yields
C* (y[t]-y[t-1]) + E * (x[t]-x[t-1]) = M[t] * (x[t]-x[t-1]);

where constants C and E will be treated as unknown variables.Figure 11 shows the complete discretized
AMPL model and Table 6 presents the behavior of the solver.

N = 100 N = 200 N = 400 N = 800

time 0.30 0.68 1.63 3.4
obj. val = -222.6832577 -222.6897622 -222.6913882 -222.6917948

infeasibility = 1.2e-12 4.1e-13 6.3e-13 3.4e-13
steps 23 25 28 30

TABLE 6. Parabolic chain

Remark 3.6. To obtain the solution with any distribution functionm(t) it is enough to change the
“mint” constraint. For example to obtain the catenary solution (m(t) = m, t ∈ [0, l]) it is enough to
replace the “mint” constraint with the following:

s.t. mint {j in 0..N-1}: M[j+1] = M[j] + m[j+1].

Table 7 shows the solver’s behavior while Figure 12 shows shapes of the parabolic and catenary chains.

param N := 200;
param d:= 5;
param ax := 0; param ay := 0;
param bx := d; param by := 0;
param l := 2 * d; param g := 9.8;
param m {1..N}, default 1;

var x {0..N};
var y {0..N};
var C;
var E;
var M {0..N};

minimize energy:
(l/N) * sum {j in 1..N-1} m[j] * g* y[j];

s.t. xleftfixed: x[0] = ax;
s.t. yleftfixed: y[0] = ay;
s.t. xrightfixed: x[N] = bx;
s.t. yrightfixed: y[N] = by;

s.t. link {j in 1..N}:
(x[j]-x[j-1])ˆ2+(y[j]-y[j-1])ˆ2)<= (l/N)ˆ2;

s.t. eqlb {j in 1..N}: M[j] * (x[j]-x[j-1])
= C* (y[j]-y[j-1]) + E * (x[j]-x[j-1]);

s.t. mint {j in 0..N-1}:
M[j+1] = M[j] + m[j+1] * (x[j+1]-x[j]);

s.t. mifixed: M[0] = 0;

let {j in 0..N} x[j] :=
(j/N) * b_x + (1-j/N) * a_x;
let {j in 0..N} y[j] :=
(j/N) * b_y + (1-j/N) * a_y;

solve;

FIGURE 11. TheAMPL modelparab.mod

FIGURE 12. Parabolic and catenary chains

N = 100 N = 200 N = 400 N = 800

time 0.34 1.10 3.74 13.18
obj. val = -223.2381140 -223.2440837 -223.2455761 -223.2459492

infeasibility = 2.9e-11 2.7e-12 1.9e-12 6.7e-12
steps 32 51 82 149

TABLE 7. Catenary chain obtained from the unified model in Figure 11

4. CONCLUSION.

Analyzing the hanging chain model we have come to several conclusions. First, discretizing the model
along the chain gives more flexibility in modeling various cases. In particular, we were able to model
the case when the mass function is uniformly distributed along the bridge span. Also, we explained the
difficulties that prevented the modeling of this situation when the problem is discretized along the horizon.

Second, when the model is discretized along the chain, it is significantly simpler to solve the correspond-
ing optimization problem using a nonlinear optimizer. In particular, the catenary problem with known mass
functionm(t) becomes a convex optimization problem. As a result, forLOQO, it takes seconds to find the
solution even for a large number of discretization points.

Third, after solving a variety of models we are convinced in the importance of a proper modeling. Prob-
lems can often be reformulated to create a simpler optimization problem and thereby reduce significantly
the solution time.

Finally, we have chosenLOQO as a solver for this paper. Other algorithms besidesLOQO could have
been used to solve the problems. Those who are interested in the behavior of different solvers for this
class of problems can refer to [3, 4], where one can find comparative studies for different solvers, and
also toNEOSwebsite http://www-neos.mcs.anl.gov/, where one can use the best optimization software for
problems formulated inAMPL.

Acknowledgements. The authors would like to thank Vladimir Aksyuk and MichaelLibman for fruitful
discussions of the paper. They are also grateful to Galina Spivak for technical assistance.

